FISHERIES RESEARCH BOARD OF CANADA

TECHNICAL REPORT NO. 376

A DIGITAL COMPUTER PROGRAM FOR CALCULATING INTEGRAL PRIMARY PRODUCTION IN VERTICALLY STRATIFIED WATERBODIES

by

EVERETT J. FEE

This is the Thirty-First FRB Technical Report from the

Fisheries Research Board of Canada

Freshwater Institute

Winnipeg, Manitoba

1973

Primary production of phytoplankton in lakes has most commonly been studied with two methods: the *in situ* bottle technique, designed to measure the integral production beneath a square meter of lake surface (Vollenweider 1969); and incubator techniques, used to study the effect of nutrients or light intensity on the rate of primary production.

Both methods have limitations. The *in situ* method is very time-consuming, usable only during a short part of the day; and it does not allow accurate predictions of production on one day from data obtained on another. Moreover, since the variations of surface light are seldom similar on different days, it is difficult to measure changes of the physiology of the photosynthetic system quantitatively.

The incubator, on the other hand, allows one to assess production potential at any time under constant light intensities, but to date no convenient way has been devised to relate these measurements to *in situ* light regimes in vertically stratified lakes in order to predict integral primary production on a routine basis.

This report presents a computer program that can be used to combine incubator measurements of photosynthetic potential with measurements of vertical light extinction in a waterbody and the distribution of surface irradiance over a day to yield an accurate estimate of daily integral primary production. It does this by simulating a continuous *in situ* exposure.

In order to use the program three kinds of data are required. The first is a curve which gives the rate of primary production per unit volume of sample as a function of irradiance. These data are obtained by measuring the rate of primary production of identical samples at four or more irradiances. For reasons given by Fee (1971), these data are best obtained with an incubator rather than *in situ*; An apparatus suitable for this purpose is described and illustrated by Fee (1973). If the waterbody is not vertically homogeneous, then a series of such curves and the depth interval over which each is applicable are required. The program assumes that there are four irradiances in the incubator. The second set of data required is the vertical extinction of light as a function of depth. These data are obtained with a submarine photometer (Vollenweider 1969). The last set of data needed is the distribution of surface light over the time interval to be integrated. A pyrheliometer is used for making these measurements. The data must be in digital format at 5-minute intervals; the program given by Fee (1971) is useful for converting strip chart output to this format. All three kinds of data are used directly in the computations; it is not necessary to fit regression lines or perform other manipulations on the raw data. Because of this, the program is very simple to use.

The computations begin by dividing the euphotic zone of the lake into 40 intervals. The per cent of surface light that penetrates to each depth is exponentially interpolated from the empirical light extinction curve. Then for the first digital value of surface irradiance, the instantaneous amount of irradiance at each depth is computed. Corresponding to the irradiance at each depth the rate of photosynthesis is interpolated from the incubator curve applicable to that depth. Numerical integration of this vertical curve gives the instantaneous rate of photosynthesis per unit area of surface. The entire procedure just outlined is repeated until depth integrals are obtained for all surface irradiance values during a day. These depth integrals are then themselves numerically integrated to give results for the entire day. A more complete description of the procedure and demonstration of its utility are given by Fee (1973a).

A program that performs the procedure just described is given in Appendix 1. The program is written FORTRAN IV and has been tested on an IBM 360/65 computer. It is not machine dependent, however, and should function without modification on any computer that compiles FORTRAN IV. Card decks of the program, including the example data set given in the appendix, will be supplied upon request.

References

Fee, E.J. 1971. Digital computer programs for estimating primary production, integrated over depth and time, in water bodies. Special Report No. 14. Center for Great Lakes Studies, University of Wisconsin-Milwaukee.

Fee, E.J. 1973. A numerical model for determining integral primary production and its application to Lake Michigan. J. Fish. Res. Bd Canada (in press).

Fee, E.J. 1973a. Modeling primary production in water bodies: a numerical approach that allows vertical inhomogeneities. J. Fish. Res. Bd Canada (in press).

Vollenweider, R.A. [ed]. 1969. A manual on methods for measuring primary production in aquatic environments including a chapter on bacteria. Blackwell Scientific Publications, Oxford and Edinburgh, 213 p.

Appendix 1

A FORTRAN IV program for computing the daily rate of photosynthetic carbon fixation beneath a square meter of lake's surface is listed below. The input data are assembled on punched cards in the format listed below. The format of the datum, i.e. either integer, real, or alpha, follows the description of the datum.

Card #1:

Col. 1-40: The irradiances of the incubator, 10 columns for each number (real). It is assumed that there are four compartments in the incubator. The irradiances are listed from the lowest irradiance to the highest. The units must be langleys (gm. cal/cm^2)/min. in the visible portion of the spectrum.

Card #2:

Col. 1-2: The number of data points on the vertical extinction of light curve (integer).

Col. 3-10: The lake name or number (alpha).

Card #3 (repeated if necessary): The depths at which transparency measurements were made; 10 columns for each number. The units must be meters (real).

Card #4 (repeated if necessary): The fraction of surface light that penetrates to each of the depths listed on the previous card(s); 10 columns for each number (real).

Card #5 (repeated if necessary):

Col. 1-40: The rates of primary production at the four incubator irradiances; 10 columns for each number (real). The units must be mgC/m^3.hr.

Col. 41-50: The depth (in meters) at which this photosynthesis curve starts to apply (real).

Col. 51-60: The depth (in meters) at which this photosynthesis curve ceases to apply (real).

Card 5 is repeated if the algae are not homogeneously distributed in the euphotic zone. A blank card or one with a negative number in columns 1-10 signals the end of this series of data.

Card #6:

Col. 1-72: A title identifying the lake and date for which the integration is being performed (alpha).

Col. 73-80: The number of data points on the surface light curve which follows on the next card (integer).

Card #7 (and following as necessary): The surface irradiance values at 5 ~~10~~-minute intervals; the units must be langleys/min. in the visible portion of the spectrum. There are 13 data points per card, each taking 6 columns (real).

To process successive data sets, that is days on which only the surface light curve changes, repeat Cards 6 and 7.

An example input data deck and the computer output given by the program are given at the end of the program listing.

```
C MAXIMUM OF 10 PHOTOSYNTHESIS VS. LIGHT FUNCTIONS TO DESCRIBE
C  VERTICAL VARIATIONS OF BIOMASS.
C MAXIMUM OF 9 IRRADIANCE LEVELS IN THE INCUBATOR (ASSUMES 4).
C MAXIMUM OF 40 POINTS ON THE VERTICAL LIGHT EXTINCTION CURVE.
C MAXIMUM OF 301 POINTS ON THE SOLAR IRRADIANCE CURVE OVER
C  A DAY.
C MAXIMUM OF 101 POINTS ON THE SIMULATED IN SITU CURVE (ASSUMES 41).
C
C THE DIMENSION STATEMENTS MUST BE CHANGED TO EXCEED THESE LIMITS.
      REAL LIGHT(40),Z(40),IRRAD(9),Z1(10),Z2(10),PROD(9,10),
     $ IZERO(301),LITE(101),P(9,101),LAMBDA
      DIMENSION ITITLE(20),ZZ(103)
      LOGICAL TOG,TOGGLE
      COMMON P,LITE,IRRAD,TOG,DELX,TOGGLE,ZZ,ITITLE
      EQUIVALENCE(IZERO(1),LIGHT(1)),(IZERO(41),Z(1)),(IZERO(81),Z1(1)),
     $ (IZERO(91),Z2(1)),(IZERO(101),PROD(1,1))
      COMMON /NUMDEP/ND,NI
C     READ IN THE IRRADIANCES OF THE INCUBATOR
C     CHANGE THE NEXT STATEMENT IF THE INCUBATOR HAS OTHER THAN
C      4 IRRADIANCE LEVELS.
      NI = 4
      READ(5,2) (IRRAD(I),I=1,NI)
2     FORMAT(8F10.0)
      WRITE(6,102)(IRRAD(I),I=1,NI)
102   FORMAT(///,' INCUBATOR IRRADIANCES (LY/MIN) ',9F10.4)
C     SET THE NUMBER OF DEPTHS TO BE USED FOR THE VERTICAL INTEGRATION.
      ND = 40
      WRITE(6,869) ND
869   FORMAT(//,' NUMPER OF DEPTH INTERVALS =',I3)
C     READ IN THE LAKE IDENTIFIER AND THE DEPTHS ON THE EXTINCTION CURVE
600   READ(5,4,END=101)  N,(ITITLE(I),I=19,20),(Z(I),I=1,N)
4     FORMAT(I2,2A4/(8F10.0))
C     READ IN THE TRANPARENCY CURVE
      READ(5,105)(LIGHT(I),I=1,N)
105   FORMAT(8F10.0)
      WRITE(6,104)  (ITITLE(I),I=19,20),(Z(I),LIGHT(I),I=1,N)
104   FORMAT(//,' LAKE ',2A4//(' Z =',F6.2,6X,'PERCENT IZERO =',F10.3))
      DO 65 I=1,N
65    LIGHT(I) = ALOG10(LIGHT(I))
      J = 0
      WRITE(6,18)
18    FORMAT(//3X,' APPLICABLE DEPTH RANGE',3X,'PHOTOSYNTHESIS '
     $,'VS. LIGHT CURVE (MG C/M**3.HR)'/1X,73('-'))
3     J = J + 1
C     READ IN THE P/I CURVE AND THE DEPTH OVER WHICH IT APPLIES.
      READ(5,1)          (PROD(I,J),I=1,NI),Z1(J),Z2(J)
C     BRANCH ON A BLANK CARD OR ONE WITH A NEGATIVE PRODUCTION.
      IF(PROD(1,J) .LE. 0.) GO TO 85
1     FORMAT(8F10.0)
      WRITE(6,1011)Z1(J),Z2(J),(PROD(I,J),I=1,NI)
```

```
1011  FORMAT(' ',F9.4,' TO ',F9.4,' M.        ', 9F10.4)
C     CONVERT RATES TO MG C/M**3.MIN
      DO 1002 I=1,NI
1002  PROD(I,J) = PROD(I,J)/60.
      GO TO 3
85    ZM = Z2(J-1)
      K = 1
      L = 1
      WRITE(6,19)
19    FORMAT('1'///' THE VERTICAL STRUCTURE OF THE PHOTOSYNTHETIC SYSTEM
     $'/'   (INTERPOLATED FROM THE INPUT DATA):'//'  I',
     $                5X,'/',4X,'DEPTH',4X,'/FRACTION IZERO /', 2X,
     $ 'PRODUCTION/LIGHT CURVE(MG C/M**3.MIN)'/1X,78('-'))
      TOG = .FALSE.
      NK = ND + 1
C     THE INTERPOLATION LOOP STARTS HERE.
      DO 10 I=1,NK
      ZED = (I-1)*ZM/ND
      IF(ZED .LE. Z(N )) GO TO 14
C     EXTRAPOLATE TO GET THE PERCENTAGE LIGHT
31    IF(TOG) GO TO 29
      TOG = .TRUE.
      SLOPE =(LIGHT(K-1) - LIGHT(K-2))/(Z(K-1)-Z(K-2))
      AINTER = LIGHT(K-2) - SLOPE*Z(K-2)
29    LITE(I) = (10.**(SLOPE*ZED + AINTER))/100.
      GO TO 10
C     INTERPOLATE THE PERCENT LIGHT VALUE
14    IF(ZED .GT. Z(K)) GO TO 11
      LITE(I) = (10.**LIGHT(K))/100.
      GO TO 12
11    IF(ZED .LT. Z(K+1)) GO TO 13
      K = K + 1
      IF(K .LE. N) GO TO 14
      GO TO 31
13    LITE(I) =(10.**(LIGHT(K+1) - (Z(K+1)-ZED)*(LIGHT(K+1)-LIGHT(K))/
     $ (Z(K+1)-Z(K))))/100.
C     INTERPOLATE THE PRODUCTIVITY CURVE APPLICABLE TO THIS DEPTH.
12    IF(ZED .GE. Z1(L) .AND. ZED .LE. Z2(L)) GO TO 15
      L = L + 1
      GO TO 12
15    DO 151 IK=1,NI
151   P(IK,I) = PROD(IK,L)
      WRITE(6,201) I,ZED,LITE(I),(P(J,I),J=1,NI)
201   FORMAT(I3,2F15.4,2X,9F11.4)
10    ZZ(I) = ZED
C     READ IN THE SURFACE LIGHT CURVE
60    READ(5,6,END=101) (ITITLE(I),I=1,18),       N,(IZERO(I),I=1,N)
6     FORMAT(18A4,I8/(13F6.3))
      WRITE(6,106) (ITITLE(I),I=1,18),  (IZERO(I),I=1,N)
106   FORMAT('1',///,' SURFACE IRRADIANCE (LY/MIN) FOR ',18A4,5X,
```

```
      $ /(10F8.3))
C     THERE MUST BE AN ODD NUMBER OF POINTS FOR THE INTEGRATION.
      IF((N/2)*2 .NE. N) GO TO 76
      N = N + 1
      IZERO(N) = 0.
76    DALY = 0.
C     CHANGE THE NEXT STATEMENT IF THE TIME INCREMENT BETWEEN DATA
C      POINTS ON THE SOLAR IRRADIANCE CURVE IS OTHER THAN 5 MINUTES.
      LAMBDA = (N-1)*5.
      TOGGLE = .FALSE.
      TOG = .FALSE.
      KK = N-3
C     DO THE INTEGRATION OVER TIME OF THE DEPTH INTEGRALS.
      DO 50 I=2,KK,2
50    DALY = DALY + 4.*DEPTH(IZERO(I),4.) + 2.*DEPTH(IZERO(I+1),2.)
      DALY = ZM*LAMBDA*(DEPTH(IZERO(1),1.) + DEPTH(IZERO(N),1.) +
     $ 4.*DEPTH(IZERO(N-1),4.) + DALY)/(9.*ND*(N-1))
      TOG = .TRUE.
      A = LAMBDA/(3.*(N-1))
C     THIS IS A DUMMY CALL TO QROD TO OBTAIN THE DEPTH PROFILE
      A = QROD(0.,0,A)
      WRITE(6,52) DALY
52    FORMAT(//' DAILY INTEGRAL PRIMARY PRODUCTION =',E15.4,
     $ ' MG C/M**2.DAY'/)
      GO TO 60
101   STOP
      END

      REAL FUNCTION DEPTH(IZ,TF)
C     SUBROUTINE DEPTH GIVES THE DEPTH INTEGRAL FOR A VALUE OF IZERO.
      REAL IZ
      COMMON /NUMDEP/ND
      IF(IZ .GT. 0.001) GO TO 1
      DEPTH = 0.
      RETURN
1     SUM = 0.
      NN = ND - 2
      DO 2 I=2,NN,2
2     SUM = SUM + 4.*QROD(IZ,I,TF) + 2.*QROD(IZ,I+1,TF)
      DEPTH = SUM + QROD(IZ,1,TF) + QROD(IZ,ND+1,TF) + 4.*QROD(IZ,ND,TF)
      RETURN
      END

      REAL FUNCTION QROD(IZ,I,FACTOR)
C     SUBROUTINE QROD GIVES THE PRODUCTION RATE AT THE IRRADIANCE ZI.
      REAL ZI,IZ,LITE(101),IRRAD(9),P(9,101)
      LOGICAL TOGGLE, TOG
      COMMON/NUMDEP/ND,NI
      COMMON P,LITE,IRRAD,TOG,DELX,TOGGLE,ZZ,ITITLE
      INTEGER ITITLE(20),IBCD(2)
```

```
      EQUIVALENCE(IBCD(1),ITITLE(19))
      REAL CURVE(103),ZZ(103)
      IF(TOG) GO TO 10
      IF(TOGGLE) GO TO 50
C     THIS IS THE FIRST TIME THROUGH THE SUBROUTINE SO INITIALIZE CURVE.
      TOGGLE = .TRUE.
      NK = ND + 1
      DO 51 J=1,NK
51    CURVE(J) = 0.
50    ZI = IZ*LITE(I)
C     TEST TO SEE WHETHER ZI IS BELOW THE LOWEST INCUBATOR IRRADIANCE
      IF(ZI .GT. IRRAD(1)) GO TO 3
      A = ZI*P(1,I)/IRRAD(1)
      GO TO 8
C     IF ZI .GT. THE HIGHEST INCUBATOR IRRADIANCE, COMPUTE INHIBITION.
3     IF(ZI .LE. IRRAD(NI)) GO TO 4
      A = (ZI - IRRAD(NI))/(.5-IRRAD(NI))
      A = P(NI,I)*(1.-A*A)
      IF(A.LT.0.)A = 0.
      GO TO 8
C     INTERPOLATE TO GET PRODUCTION RATE..
4     J = 2
6     IF(ZI.LE.IRRAD(J)) GO TO 5
      J = J + 1
      GO TO 6
5     A = ((P(J,I)-P(J-1,I))/(IRRAD(J)-IRRAD(J-1)))*(ZI-IRRAD(J))+P(J,I)
8     QROD = A
      CURVE(I) = CURVE(I) + A*FACTOR
      RETURN
10    DO 12 I=1,NK
12    CURVE(I) = FACTOR * CURVE(I)
      WRITE(6,11)(I,ZZ(I),CURVE(I),I=1,NK)
11    FORMAT(//' THE PHOTOSYNTHESIS/DEPTH CURVE IS:'//'   I',20X,'DEPTH',
     $7X,'MG C/M**3.DAY'/1X,48('-')/(I4,5X,2F20.4))
      QROD = 0.
      RETURN
      END
```

THIS IS AN EXAMPLE DATA SET THAT MAY BE USED TO TEST THE PROGRAM.
THE FIRST LINE IS NOT A DATA CARD BUT RATHER INDICATES THE
COLUMN IN WHICH THE DATA ARE PUNCHED ON THE CARDS.

```
12345678901234567890123456789012345678901234567890123456789012345678901234567890

0.0041    0.0098    0.0322    0.1120
29 228    21.7.72
0.        .25       .5        1.        2.        2.5       3.        4.
5.        6.        7.        8.        9.        10.       11.       12.
13.       14.       15.       16.       17.       18.       19.       20.
21.       22.       23.       24.       25.
100.      94.       88.       80.       62.       55.       50.       44.
36.       28.       22.       18.       14.       12.       10.       8.4
7.0       5.8       4.6       3.8       3.2       3.0       2.6       2.0
1.8       1.6       1.4       1.2       1.0
0.362     0.289     0.497     0.363     0.        .5
0.140     0.279     0.452     0.614     .5        2.
.210      .430      .465      .424      2.        3.5
.180      .354      .466      .412      3.5       7.5
.196      .415      .505      .448      7.5       15.
.209      .371      .378      .281      15.       20.
-100.
 JULY 21, 1972. L 228(TEGGAU) TEGGAU MET SITE.                                49
 0.0    0.084 0.156 0.204 0.099 0.145 0.253 0.098 0.003 0.005 0.054 0.019 0.047
 0.144 0.235 0.248 0.174 0.131 0.085 0.094 0.019 0.002 0.006 0.005 0.032 0.008
 0.005 0.009 0.002 0.007 0.008 0.073 0.172 0.125 0.015 0.089 0.137 0.125 0.130
 0.120 0.101 0.120 0.125 0.117 0.164 0.144 0.181 0.040 0.026
```

THE OUTPUT GENERATED BY THE PROGRAM FOLLOWS ON THE NEXT PAGES.

INCUBATOR IRRADIANCES (LY/MIN) 0.0041 0.0098 0.0322 0.1120

NUMBER OF DEPTH INTERVALS = 40

LAKE 228

Z = 0.0	PERCENT IZERO =	100.000
Z = 0.25	PERCENT IZERO =	94.000
Z = 0.50	PERCENT IZERO =	88.000
Z = 1.00	PERCENT IZERO =	80.000
Z = 2.00	PERCENT IZERO =	62.000
Z = 2.50	PERCENT IZERO =	55.000
Z = 3.00	PERCENT IZERO =	50.000
Z = 4.00	PERCENT IZERO =	44.000
Z = 5.00	PERCENT IZERO =	36.000
Z = 6.00	PERCENT IZERO =	28.000
Z = 7.00	PERCENT IZERO =	22.000
Z = 8.00	PERCENT IZERO =	18.000
Z = 9.00	PERCENT IZERO =	14.000
Z = 10.00	PERCENT IZERO =	12.000
Z = 11.00	PERCENT IZERO =	10.000
Z = 12.00	PERCENT IZERO =	8.400
Z = 13.00	PERCENT IZERO =	7.000
Z = 14.00	PERCENT IZERO =	5.800
Z = 15.00	PERCENT IZERO =	4.600
Z = 16.00	PERCENT IZERO =	3.800
Z = 17.00	PERCENT IZERO =	3.200
Z = 18.00	PERCENT IZERO =	3.000
Z = 19.00	PERCENT IZERO =	2.600
Z = 20.00	PERCENT IZERO =	2.000
Z = 21.00	PERCENT IZERO =	1.800
Z = 22.00	PERCENT IZERO =	1.600
Z = 23.00	PERCENT IZERO =	1.400
Z = 24.00	PERCENT IZERO =	1.200
Z = 25.00	PERCENT IZERO =	1.000

APPLICABLE DEPTH RANGE	PHOTOSYNTHESIS VS. LIGHT CURVE (MG C/M**3.HR)			
0.0 TO 0.5000 M.	0.3620	0.2890	0.4970	0.3630
0.5000 TO 2.0000 M.	0.1400	0.2790	0.4520	0.6140
2.0000 TO 3.5000 M.	0.2100	0.4300	0.4650	0.4240
3.5000 TO 7.5000 M.	0.1800	0.3540	0.4660	0.4120
7.5000 TO 15.0000 M.	0.1960	0.4150	0.5050	0.4480
15.0000 TO 20.0000 M.	0.2090	0.3710	0.3780	0.2810

THE VERTICAL STRUCTURE OF THE PHOTOSYNTHETIC SYSTEM
(INTERPOLATED FROM THE INPUT DATA):

I /	DEPTH	/FRACTION IZERO /	PRODUCTION/LIGHT CURVE(MG C/M**3.MIN)			
1	0.0	1.0000	0.0060	0.0048	0.0083	0.0060
2	0.5000	0.8800	0.0060	0.0048	0.0083	0.0060
3	1.0000	0.8000	0.0023	0.0046	0.0075	0.0102
4	1.5000	0.7043	0.0023	0.0046	0.0075	0.0102
5	2.0000	0.6200	0.0023	0.0046	0.0075	0.0102
6	2.5000	0.5500	0.0035	0.0072	0.0077	0.0071
7	3.0000	0.5000	0.0035	0.0072	0.0077	0.0071
8	3.5000	0.4690	0.0035	0.0072	0.0077	0.0071
9	4.0000	0.4400	0.0030	0.0059	0.0078	0.0069
10	4.5000	0.3980	0.0030	0.0059	0.0078	0.0069
11	5.0000	0.3600	0.0030	0.0059	0.0078	0.0069
12	5.5000	0.3175	0.0030	0.0059	0.0078	0.0069
13	6.0000	0.2800	0.0030	0.0059	0.0078	0.0069
14	6.5000	0.2482	0.0030	0.0059	0.0078	0.0069
15	7.0000	0.2200	0.0030	0.0059	0.0078	0.0069
16	7.5000	0.1990	0.0030	0.0059	0.0078	0.0069
17	8.0000	0.1800	0.0033	0.0069	0.0084	0.0075
18	8.5000	0.1587	0.0033	0.0069	0.0084	0.0075
19	9.0000	0.1400	0.0033	0.0069	0.0084	0.0075
20	9.5000	0.1296	0.0033	0.0069	0.0084	0.0075
21	10.0000	0.1200	0.0033	0.0069	0.0084	0.0075
22	10.5000	0.1095	0.0033	0.0069	0.0084	0.0075
23	11.0000	0.1000	0.0033	0.0069	0.0084	0.0075
24	11.5000	0.0917	0.0033	0.0069	0.0084	0.0075
25	12.0000	0.0840	0.0033	0.0069	0.0084	0.0075
26	12.5000	0.0767	0.0033	0.0069	0.0084	0.0075
27	13.0000	0.0700	0.0033	0.0069	0.0084	0.0075
28	13.5000	0.0637	0.0033	0.0069	0.0084	0.0075
29	14.0000	0.0580	0.0033	0.0069	0.0084	0.0075
30	14.5000	0.0517	0.0033	0.0069	0.0084	0.0075
31	15.0000	0.0460	0.0033	0.0069	0.0084	0.0075
32	15.5000	0.0418	0.0035	0.0062	0.0063	0.0047
33	16.0000	0.0380	0.0035	0.0062	0.0063	0.0047
34	16.5000	0.0349	0.0035	0.0062	0.0063	0.0047
35	17.0000	0.0320	0.0035	0.0062	0.0063	0.0047
36	17.5000	0.0310	0.0035	0.0062	0.0063	0.0047
37	18.0000	0.0300	0.0035	0.0062	0.0063	0.0047
38	18.5000	0.0279	0.0035	0.0062	0.0063	0.0047
39	19.0000	0.0260	0.0035	0.0062	0.0063	0.0047
40	19.5000	0.0228	0.0035	0.0062	0.0063	0.0047
41	20.0000	0.0200	0.0035	0.0062	0.0063	0.0047

SURFACE IRRADIANCE (LY/MIN) FOR JULY 21, 1972. L 228(TEGGAU) TEGGAU MET SITE.

0.0	0.084	0.156	0.204	0.099	0.145	0.253	0.098	0.003	0.005
0.054	0.019	0.047	0.144	0.235	0.248	0.174	0.131	0.085	0.094
0.019	0.002	0.006	0.005	0.032	0.008	0.005	0.009	0.002	0.007
0.008	0.073	0.172	0.125	0.015	0.089	0.137	0.125	0.130	0.120
0.101	0.120	0.125	0.117	0.164	0.144	0.181	0.040	0.026	

THE PHOTOSYNTHESIS/DEPTH CURVE IS:

I	DEPTH	MG C/M**3.DAY
1	0.0	1.4277
2	0.5000	1.4497
3	1.0000	1.7872
4	1.5000	1.7283
5	2.0000	1.6677
6	2.5000	1.4882
7	3.0000	1.4787
8	3.5000	1.4699
9	4.0000	1.4077
10	4.5000	1.3979
11	5.0000	1.3868
12	5.5000	1.3678
13	6.0000	1.3451
14	6.5000	1.3183
15	7.0000	1.2810
16	7.5000	1.2456
17	8.0000	1.3492
18	8.5000	1.3083
19	9.0000	1.2700
20	9.5000	1.2470
21	10.0000	1.2222
22	10.5000	1.1916
23	11.0000	1.1573
24	11.5000	1.1198
25	12.0000	1.0846
26	12.5000	1.0411
27	13.0000	0.9889
28	13.5000	0.9305
29	14.0000	0.8721
30	14.5000	0.8011
31	15.0000	0.7339
32	15.5000	0.6736
33	16.0000	0.6301
34	16.5000	0.5917
35	17.0000	0.5538
36	17.5000	0.5394
37	18.0000	0.5252
38	18.5000	0.4945
39	19.0000	0.4642
40	19.5000	0.4126
41	20.0000	0.3656

DAILY INTEGRAL PRIMARY PRODUCTION = 0.2191E 02 MG C/M**2.DAY